
Syntax

streg
[

varlist
] [

if
] [

in
] [

, options
]

options Description

Model

noconstant suppress constant term
distribution(exponential) exponential survival distribution
distribution(gompertz) Gompertz survival distribution
distribution(loglogistic) loglogistic survival distribution
distribution(llogistic) synonym for distribution(loglogistic)
distribution(weibull) Weibull survival distribution
distribution(lognormal) lognormal survival distribution
distribution(lnormal) synonym for distribution(lognormal)
distribution(gamma) generalized gamma survival distribution
frailty(gamma) gamma frailty distribution
frailty(invgaussian) inverse-Gaussian distribution
time use accelerated failure-time metric

Model 2

strata(varname) strata ID variable
offset(varname) include varname in model with coefficient constrained to 1
shared(varname) shared frailty ID variable
ancillary(varlist) use varlist to model the first ancillary parameter
anc2(varlist) use varlist to model the second ancillary parameter
constraints(constraints) apply specified linear constraints
collinear keep collinear variables

SE/Robust

vce(vcetype) vcetype may be oim, robust, cluster clustvar, opg,
bootstrap, or jackknife

2 streg — Parametric survival models

Reporting

level(#) set confidence level; default is level(95)

nohr do not report hazard ratios
tr report time ratios
noshow do not show st setting information
noheader suppress header from coefficient table
nolrtest do not perform likelihood-ratio test
nocnsreport do not display constraints
display options control column formats, row spacing, line width, display of omitted

variables and base and empty cells, and factor-variable labeling

Maximization

maximize options control the maximization process; seldom used

coeflegend display legend instead of statistics

You must stset your data before using streg; see [ST] stset.
varlist may contain factor variables; see [U] 11.4.3 Factor variables.
bootstrap, by, fp, jackknife, mfp, mi estimate, nestreg, statsby, stepwise, and svy are allowed; see

[U] 11.1.10 Prefix commands.
vce(bootstrap) and vce(jackknife) are not allowed with the mi estimate prefix; see [MI] mi estimate.
shared(), vce(), and noheader are not allowed with the svy prefix; see [SVY] svy.
fweights, iweights, and pweights may be specified using stset; see [ST] stset. However, weights may not be

specified if you are using the bootstrap prefix with the streg command.
coeflegend does not appear in the dialog box.
See [U] 20 Estimation and postestimation commands for more capabilities of estimation commands.

Menu
Statistics > Survival analysis > Regression models > Parametric survival models

Description
streg performs maximum likelihood estimation for parametric regression survival-time models.

streg can be used with single- or multiple-record or single- or multiple-failure st data. Survival
models currently supported are exponential, Weibull, Gompertz, lognormal, loglogistic, and generalized
gamma. Parametric frailty models and shared-frailty models are also fit using streg.

Also see [ST] stcox for proportional hazards models.

Options

� � �
Model �

noconstant; see [R] estimation options.

distribution(distname) specifies the survival model to be fit. A specified distribution() is
remembered from one estimation to the next when distribution() is not specified.

For instance, typing streg x1 x2, distribution(weibull) fits a Weibull model. Subsequently,
you do not need to specify distribution(weibull) to fit other Weibull regression models.

All Stata estimation commands, including streg, redisplay results when you type the command
name without arguments. To fit a model with no explanatory variables, type streg, distribu-
tion(distname). . . .

frailty(gamma | invgaussian) specifies the assumed distribution of the frailty, or heterogeneity.
The estimation results, in addition to the standard parameter estimates, will contain an estimate of
the variance of the frailties and a likelihood-ratio test of the null hypothesis that this variance is
zero. When this null hypothesis is true, the model reduces to the model with frailty(distname)
not specified.

A specified frailty() is remembered from one estimation to the next when distribution()
is not specified. When you specify distribution(), the previously remembered specification of
frailty() is forgotten.

time specifies that the model be fit in the accelerated failure-time metric rather than in the log
relative-hazard metric. This option is valid only for the exponential and Weibull models because
these are the only models that have both a proportional hazards and an accelerated failure-time
parameterization. Regardless of metric, the likelihood function is the same, and models are equally
appropriate viewed in either metric; it is just a matter of changing the interpretation.

time must be specified at estimation.

� � �
Model 2 �

strata(varname) specifies the stratification ID variable. Observations with equal values of the variable
are assumed to be in the same stratum. Stratified estimates (with equal coefficients across strata
but intercepts and ancillary parameters unique to each stratum) are then obtained. This option is
not available if frailty(distname) is specified.

offset(varname); see [R] estimation options.

shared(varname) is valid with frailty() and specifies a variable defining those groups over which
the frailty is shared, analogous to a random-effects model for panel data where varname defines the
panels. frailty() specified without shared() treats the frailties as occurring at the observation
level.

A specified shared() is remembered from one estimation to the next when distribution()
is not specified. When you specify distribution(), the previously remembered specification of
shared() is forgotten.

shared() may not be used with distribution(gamma), vce(robust), vce(cluster clustvar),
vce(opg), the svy prefix, or in the presence of delayed entries or gaps.

If shared() is specified without frailty() and there is no remembered frailty() from the
previous estimation, frailty(gamma) is assumed to provide behavior analogous to stcox; see
[ST] stcox.

ancillary(varlist) specifies that the ancillary parameter for the Weibull, lognormal, Gompertz, and
loglogistic distributions and that the first ancillary parameter (sigma) of the generalized log-gamma
distribution be estimated as a linear combination of varlist. This option may not be used with
frailty(distname).

When an ancillary parameter is constrained to be strictly positive, the logarithm of the ancillary
parameter is modeled as a linear combination of varlist.

http://www.stata.com/manuals13/u11.pdf#u11.3Namingconventions
http://www.stata.com/manuals13/u11.pdf#u11.3Namingconventions

anc2(varlist) specifies that the second ancillary parameter (kappa) for the generalized log-gamma
distribution be estimated as a linear combination of varlist. This option may not be used with
frailty(distname).

constraints(constraints), collinear; see [R] estimation options.

� � �
SE/Robust �

vce(vcetype) specifies the type of standard error reported, which includes types that are derived from
asymptotic theory (oim, opg), that are robust to some kinds of misspecification (robust), that
allow for intragroup correlation (cluster clustvar), and that use bootstrap or jackknife methods
(bootstrap, jackknife); see [R] vce option.

� � �
Reporting �

level(#); see [R] estimation options.

nohr, which may be specified at estimation or upon redisplaying results, specifies that coefficients
rather than exponentiated coefficients be displayed, that is, that coefficients rather than hazard ratios
be displayed. This option affects only how coefficients are displayed, not how they are estimated.

This option is valid only for models with a natural proportional-hazards parameterization: exponen-
tial, Weibull, and Gompertz. These three models, by default, report hazard ratios (exponentiated
coefficients).

tr specifies that exponentiated coefficients, which are interpreted as time ratios, be displayed. tr is
appropriate only for the loglogistic, lognormal, and gamma models, or for the exponential and
Weibull models when fit in the accelerated failure-time metric.

tr may be specified at estimation or upon replay.

noshow prevents streg from showing the key st variables. This option is rarely used because most
people type stset, show or stset, noshow to set once and for all whether they want to see
these variables mentioned at the top of the output of every st command; see [ST] stset.

noheader suppresses the output header, either at estimation or upon replay.

nolrtest is valid only with frailty models, in which case it suppresses the likelihood-ratio test for
significant frailty.

nocnsreport; see [R] estimation options.

display options: noomitted, vsquish, noemptycells, baselevels, allbaselevels, nofvla-
bel, fvwrap(#), fvwrapon(style), cformat(% fmt), pformat(% fmt), sformat(% fmt), and
nolstretch; see [R] estimation options.

� � �
Maximization �

maximize options: difficult, technique(algorithm spec), iterate(#),
[
no
]
log, trace,

gradient, showstep, hessian, showtolerance, tolerance(#), ltolerance(#),
nrtolerance(#), nonrtolerance, and from(init specs); see [R] maximize. These options are
seldom used.

Setting the optimization type to technique(bhhh) resets the default vcetype to vce(opg).

The following option is available with streg but is not shown in the dialog box:

coeflegend; see [R] estimation options.

http://www.stata.com/manuals13/u11.pdf#u11.4varlists
http://www.stata.com/manuals13/restimationoptions.pdf#restimationoptions
http://www.stata.com/manuals13/rvce_option.pdf#rvce_option
http://www.stata.com/manuals13/restimationoptions.pdf#restimationoptions
http://www.stata.com/manuals13/ststset.pdf#ststset
http://www.stata.com/manuals13/restimationoptions.pdf#restimationoptions
http://www.stata.com/manuals13/d.pdf#dformat
http://www.stata.com/manuals13/restimationoptions.pdf#restimationoptions

Remarks and examples stata.com

Remarks are presented under the following headings:

Introduction
Distributions

Weibull and exponential models
Gompertz model
Lognormal and loglogistic models
Generalized gamma model

Examples
Parameterization of ancillary parameters
Stratified estimation
(Unshared-) frailty models
Shared-frailty models

Introduction

What follows is a brief summary of what you can do with streg. For a complete tutorial, see
Cleves et al. (2010), which devotes four chapters to this topic.

Two often-used models for adjusting survivor functions for the effects of covariates are the
accelerated failure-time (AFT) model and the multiplicative or proportional hazards (PH) model. In
the AFT model, the natural logarithm of the survival time, log t, is expressed as a linear function of
the covariates, yielding the linear model

log tj = xjβ + zj

where xj is a vector of covariates, β is a vector of regression coefficients, and zj is the error with
density f(). The distributional form of the error term determines the regression model. If we let f()
be the normal density, the lognormal regression model is obtained. Similarly, by letting f() be the
logistic density, the loglogistic regression is obtained. Setting f() equal to the extreme-value density
yields the exponential and the Weibull regression models.

The effect of the AFT model is to change the time scale by a factor of exp(−xjβ). Depending
on whether this factor is greater or less than 1, time is either accelerated or decelerated (degraded).
That is, if a subject at baseline experiences a probability of survival past time t equal to S(t), then
a subject with covariates xj would have probability of survival past time t equal to S() evaluated at
the point exp(−xjβ)t, instead. Thus accelerated failure time does not imply a positive acceleration
of time with the increase of a covariate but instead implies a deceleration of time or, equivalently, an
increase in the expected waiting time for failure.

In the PH model, the concomitant covariates have a multiplicative effect on the hazard function

h(tj) = h0(t)g(xj)

for some h0(t), and for g(xj), a nonnegative function of the covariates. A popular choice, and the
one adopted here, is to let g(xj) = exp(xjβ). The function h0(t) may either be left unspecified,
yielding the Cox proportional hazards model (see [ST] stcox), or take a specific parametric form.
For the streg command, h0(t) is assumed to be parametric. Three regression models are currently
implemented as PH models: the exponential, Weibull, and Gompertz models. The exponential and
Weibull models are implemented as both AFT and PH models, and the Gompertz model is implemented
only in the PH metric.

http://stata.com
http://www.stata.com/manuals13/ststcox.pdf#ststcox

The above model allows for the presence of an intercept term, β0, within xjβ. Thus what is
commonly referred to as the baseline hazard function—the hazard when all covariates are zero—is
actually equal to h0(t) exp(β0). That is, the intercept term serves to scale the baseline hazard. Of
course, specifying noconstant suppresses the intercept or equivalently constrains β0 to equal zero.

streg is suitable only for data that have been stset. By stsetting your data, you define the
variables t0, t, and d, which serve as the trivariate response variable (t0, t, d). Each response
corresponds to a period under observation, (t0, t], resulting in either failure (d = 1) or right-censoring
(d = 0) at time t. As a result, streg is appropriate for data exhibiting delayed entry, gaps, time-varying
covariates, and even multiple-failure data.

Distributions
Six parametric survival distributions are currently supported by streg. The parameterization and

ancillary parameters for each distribution are summarized in table 1:

Table 1. Parametric survival distributions supported by streg

Ancillary
Distribution Metric Survivor function Parameterization parameters

Exponential PH exp(−λjtj) λj = exp(xjβ)
Exponential AFT exp(−λjtj) λj = exp(−xjβ)
Weibull PH exp(−λjtpj ) λj = exp(xjβ) p

Weibull AFT exp(−λjtpj ) λj = exp(−pxjβ) p

Gompertz PH exp{−λjγ−1(eγtj − 1)} λj = exp(xjβ) γ

Lognormal AFT 1− Φ
{

log(tj)−µj
σ

}
µj = xjβ σ

Loglogistic AFT {1 + (λjtj)1/γ}−1 λj = exp(−xjβ) γ

Generalized gamma
if κ > 0 AFT 1− I

(
γ, u

)
µj = xjβ σ, κ

if κ = 0 AFT 1− Φ(z) µj = xjβ σ, κ
if κ < 0 AFT I

(
γ, u

)
µj = xjβ σ, κ

where PH = proportional hazards, AFT = accelerated failure time, and Φ(z) is the standard normal
cumulative distribution. For the generalized gamma, γ = |κ|−2, u = γexp(|κ|z), I(a, x) is the
incomplete gamma function, and z = sign(κ){log(tj)− µj}/σ.

Plotted in figure 1 are example hazard functions for five of the six distributions. The exponential
hazard (not separately plotted) is a special case of the Weibull hazard when the Weibull ancillary
parameter p = 1. The generalized gamma (not plotted) is extremely flexible and therefore can take
many shapes.

Figure 1. Example plots of hazard functions

Weibull and exponential models

The Weibull and exponential models are parameterized as both PH and AFT models. The Weibull
distribution is suitable for modeling data with monotone hazard rates that either increase or decrease
exponentially with time, whereas the exponential distribution is suitable for modeling data with
constant hazard (see figure 1).

For the PH model, h0(t) = 1 for exponential regression, and h0(t) = p tp−1 for Weibull regression,
where p is the shape parameter to be estimated from the data. Some authors refer not to p but to
σ = 1/p.

The AFT model is written as
log(tj) = xjβ∗ + zj

where zj has an extreme-value distribution scaled by σ. Let β be the vector of regression coefficients
derived from the PH model so that β∗ = −σβ. This relationship holds only if the ancillary parameter,
p, is a constant; it does not hold when the ancillary parameter is parameterized in terms of covariates.

streg uses, by default, for the exponential and Weibull models, the proportional-hazards metric
simply because it eases comparison with those results produced by stcox (see [ST] stcox). You can,
however, specify the time option to choose the accelerated failure-time parameterization.

The Weibull hazard and survivor functions are

h(t) = pλtp−1

http://www.stata.com/manuals13/ststcox.pdf#ststcox

S(t) = exp(−λtp)

where λ is parameterized as described in table 1. If p = 1, these functions reduce to those of the
exponential.

Gompertz model

The Gompertz regression is parameterized only as a PH model. First described in 1825, this
model has been extensively used by medical researchers and biologists modeling mortality data. The
Gompertz distribution implemented is the two-parameter function as described in Lee and Wang (2003),
with the following hazard and survivor functions:

h(t) = λ exp(γt)

S(t) = exp{−λγ−1(eγt − 1)}

The model is implemented by parameterizing λj = exp(xjβ), implying that h0(t) = exp(γt),
where γ is an ancillary parameter to be estimated from the data.

This distribution is suitable for modeling data with monotone hazard rates that either increase or
decrease exponentially with time (see figure 1).

When γ is positive, the hazard function increases with time; when γ is negative, the hazard
function decreases with time; and when γ is zero, the hazard function is equal to λ for all t, so the
model reduces to an exponential.

Some recent survival analysis texts, such as Klein and Moeschberger (2003), restrict γ to be strictly
positive. If γ < 0, then as t goes to infinity, the survivor function, S(t), exponentially decreases
to a nonzero constant, implying that there is a nonzero probability of never failing (living forever).
That is, there is always a nonzero hazard rate, yet it decreases exponentially. By restricting γ to be
positive, we know that the survivor function always goes to zero as t tends to infinity.

Although the above argument may be desirable from a mathematical perspective, in Stata’s
implementation, we took the more traditional approach of not restricting γ. We did this because, in
survival studies, subjects are not monitored forever—there is a date when the study ends, and in many
investigations, specifically in medical research, an exponentially decreasing hazard rate is clinically
appealing.

Lognormal and loglogistic models

The lognormal and loglogistic models are implemented only in the AFT form. These two distributions
are similar and tend to produce comparable results. For the lognormal distribution, the natural logarithm
of time follows a normal distribution; for the loglogistic distribution, the natural logarithm of time
follows a logistic distribution.

The lognormal survivor and density functions are

S(t) = 1− Φ
{

log(t)− µ
σ

}

f (t) =
1

tσ
√

2π
exp
[
−1
2σ2

{
log(t)− µ

}2
]

where Φ(z) is the standard normal cumulative distribution function.



streg — Parametric survival models 9

The lognormal regression is implemented by setting µj = xjβ and treating the standard deviation,
σ, as an ancillary parameter to be estimated from the data.

The loglogistic regression is obtained if zj has a logistic density. The loglogistic survivor and
density functions are

S(t) = {1 + (λt)1/γ}−1

f (t) =
λ1/γt1/γ−1

γ
{

1 + (λt)1/γ}2

This model is implemented by parameterizing λj = exp(−xjβ) and treating the scale parameter
γ as an ancillary parameter to be estimated from the data.

Unlike the exponential, Weibull, and Gompertz distributions, the lognormal and the loglogistic
distributions are indicated for data exhibiting nonmonotonic hazard rates, specifically initially increasing
and then decreasing rates (figure 1).

Thus far we have considered the exponential, Weibull, lognormal, and loglogistic models. These
models are sufficiently flexible for many datasets, but further flexibility can be obtained with the
generalized gamma model, described below. Alternatively, you might consider using a Royston–
Parmar model (Royston and Parmar 2002; Lambert and Royston 2009). Royston–Parmar models are
highly flexible alternatives to the exponential, Weibull, lognormal, and loglogistic models that allow
extension from proportional hazards to proportional odds and to scaled probit models. Additional
flexibility can be obtained with restricted cubic spline functions as alternatives to the linear functions
of log time considered in Introduction. See Royston and Lambert (2011) for a thorough treatment of
this topic.

Generalized gamma model

The generalized gamma model is implemented only in the AFT form. The three-parameter generalized
gamma survivor and density functions are

S(t) =


1− I

(
γ, u

)
, if κ > 0

1− Φ(z), if κ = 0
I
(
γ, u

)
, if κ < 0

f(t) =

{ γγ

σt
√
γΓ(γ) exp

(
z
√
γ − u), if κ 6= 0

1
σt
√

2π
exp(−z2/2), if κ = 0

where γ = |κ|−2, z = sign(κ){log(t) − µ}/σ, u = γ exp(|κ|z), Φ(z) is the standard normal
cumulative distribution function, and I(a, x) is the incomplete gamma function. See the gammap(a,x)
entry in [D] functions to see how the incomplete gamma function is implemented in Stata.

This model is implemented by parameterizing µj = xjβ and treating the parameters κ and σ as
ancillary parameters to be estimated from the data.

The hazard function of the generalized gamma distribution is extremely flexible, allowing for many
possible shapes, including as special cases the Weibull distribution when κ = 1, the exponential when
κ = 1 and σ = 1, and the lognormal distribution when κ = 0. The generalized gamma model is,
therefore, commonly used for evaluating and selecting an appropriate parametric model for the data.
The Wald or likelihood-ratio test can be used to test the hypotheses that κ = 1 or that κ = 0.

http://www.stata.com/manuals13/dfunctions.pdf#dfunctionsDescriptiongammap()
http://www.stata.com/manuals13/dfunctions.pdf#dfunctions

Examples

Example 1

The Weibull distribution provides a good illustration of streg because this distribution is param-
eterized as both AFT and PH and serves to compare and contrast the two approaches.

We wish to analyze an experiment testing the ability of emergency generators with new-style
bearings to withstand overloads. This dataset is described in [ST] stcox. This time, we wish to fit a
Weibull model:

. use http://www.stata-press.com/data/r13/kva
(Generator experiment)

. streg load bearings, distribution(weibull)

failure _d: 1 (meaning all fail)
analysis time _t: failtime

Fitting constant-only model:

Iteration 0: log likelihood = -13.666193
Iteration 1: log likelihood = -9.7427276
Iteration 2: log likelihood = -9.4421169
Iteration 3: log likelihood = -9.4408287
Iteration 4: log likelihood = -9.4408286

Fitting full model:

Iteration 0: log likelihood = -9.4408286
Iteration 1: log likelihood = -2.078323
Iteration 2: log likelihood = 5.2226016
Iteration 3: log likelihood = 5.6745808
Iteration 4: log likelihood = 5.6934031
Iteration 5: log likelihood = 5.6934189
Iteration 6: log likelihood = 5.6934189

Weibull regression -- log relative-hazard form

No. of subjects = 12 Number of obs = 12
No. of failures = 12
Time at risk = 896

LR chi2(2) = 30.27
Log likelihood = 5.6934189 Prob > chi2 = 0.0000

_t Haz. Ratio Std. Err. z P>|z| [95% Conf. Interval]

load 1.599315 .1883807 3.99 0.000 1.269616 2.014631
bearings .1887995 .1312109 -2.40 0.016 .0483546 .7371644

_cons 2.51e-20 2.66e-19 -4.26 0.000 2.35e-29 2.68e-11

/ln_p 2.051552 .2317074 8.85 0.000 1.597414 2.505691

p 7.779969 1.802677 4.940241 12.25202
1/p .1285352 .0297826 .0816192 .2024193

Because we did not specify otherwise, the estimation took place in the hazard metric, which is
the default for distribution(weibull). The estimates are directly comparable to those produced
by stcox: stcox estimated a hazard ratio of 1.526 for load and 0.0636 for bearings.

However, we estimated the baseline hazard function as well, assuming that it is Weibull. The
estimates are the full maximum-likelihood estimates. The shape parameter is fit as ln p, but streg
then reports p and 1/p = σ so that you can think about the parameter however you wish.

http://www.stata.com/manuals13/ststcox.pdf#ststcox

We find that p is greater than 1, which means that the hazard of failure increases with time and,
here, increases dramatically. After 100 hours, the bearings are more than 1 million times more likely
to fail per second than after 10 hours (or, to be precise, (100/10)7.78−1). From our knowledge of
generators, we would expect this; it is the accumulation of heat due to friction that causes bearings
to expand and seize.

Technical note
Regression results are often presented in a metric other than the natural regression coefficients, that

is, as hazard ratios, relative risk ratios, odds ratios, etc. In those cases, standard errors are calculated
using the delta method.

However, the Z test and p-values given are calculated from the natural regression coefficients
and standard errors. Although a test based on, say, a hazard ratio and its standard error would be
asymptotically equivalent to that based on a regression coefficient, in real samples a hazard ratio will
tend to have a more skewed distribution because it is an exponentiated regression coefficient. Also,
it is more natural to think of these tests as testing whether a regression coefficient is nonzero, rather
than testing whether a transformed regression coefficient is unequal to some nonzero value (one for
a hazard ratio).

Finally, the confidence intervals given are obtained by transforming the endpoints of the cor-
responding confidence interval for the untransformed regression coefficient. This ensures that, say,
strictly positive quantities such as hazard ratios have confidence intervals that do not overlap zero.

Example 2

The previous estimation took place in the PH metric, and exponentiated coefficients—hazard
ratios—were reported. If we want to see the unexponentiated coefficients, we could redisplay results
and specify the nohr option:

. streg, nohr

Weibull regression -- log relative-hazard form

No. of subjects = 12 Number of obs = 12
No. of failures = 12
Time at risk = 896

LR chi2(2) = 30.27
Log likelihood = 5.6934189 Prob > chi2 = 0.0000

_t Coef. Std. Err. z P>|z| [95% Conf. Interval]

load .4695753 .1177884 3.99 0.000 .2387143 .7004363
bearings -1.667069 .6949745 -2.40 0.016 -3.029194 -.3049443

_cons -45.13191 10.60663 -4.26 0.000 -65.92053 -24.34329

/ln_p 2.051552 .2317074 8.85 0.000 1.597414 2.505691

p 7.779969 1.802677 4.940241 12.25202
1/p .1285352 .0297826 .0816192 .2024193

Example 3

We could just as well have fit this model in the AFT metric:

. streg load bearings, d(weibull) time nolog

failure _d: 1 (meaning all fail)
analysis time _t: failtime

Weibull regression -- accelerated failure-time form

No. of subjects = 12 Number of obs = 12
No. of failures = 12
Time at risk = 896

LR chi2(2) = 30.27
Log likelihood = 5.6934189 Prob > chi2 = 0.0000

_t Coef. Std. Err. z P>|z| [95% Conf. Interval]

load -.060357 .0062214 -9.70 0.000 -.0725507 -.0481632
bearings .2142771 .0746451 2.87 0.004 .0679753 .3605789

_cons 5.80104 .1752301 33.11 0.000 5.457595 6.144485

/ln_p 2.051552 .2317074 8.85 0.000 1.597414 2.505691

p 7.779969 1.802677 4.940241 12.25202
1/p .1285352 .0297826 .0816192 .2024193

This is the same model we previously fit, but it is presented in a different metric. Calling the
previous coefficients b, these coefficients are −σb = −b/p. For instance, in the previous example,
the coefficient on load was reported as 0.4695753, and −0.4695753/7.779969 = −0.06035696. d()
is a convenient shorthand for distribution().

Example 4

streg may also be applied to more complicated data. Below we have multiple records per subject
on a failure that can occur repeatedly:

. use http://www.stata-press.com/data/r13/mfail3

. stdescribe

per subject
Category total mean min median max

no. of subjects 926
no. of records 1734 1.87257 1 2 4

(first) entry time 0 0 0 0
(final) exit time 470.6857 1 477 960

subjects with gap 6
time on gap if gap 411 68.5 16 57.5 133
time at risk 435444 470.2419 1 477 960

failures 808 .8725702 0 1 3

In this dataset, subjects have up to four records (most have two) and have up to three failures (most
have one) and, although you cannot tell from the above output, the data have time-varying covariates,
as well. There are even six subjects with gaps in their histories, meaning that, for a while, they went
unobserved. Although we could estimate in the AFT metric, it is easier to interpret results in the PH
metric (or the log relative-hazard metric, as it is also known):

. streg x1 x2, d(weibull) vce(robust)

Fitting constant-only model:

Iteration 0: log pseudolikelihood = -1398.2504
Iteration 1: log pseudolikelihood = -1382.8224
Iteration 2: log pseudolikelihood = -1382.7457
Iteration 3: log pseudolikelihood = -1382.7457

Fitting full model:

Iteration 0: log pseudolikelihood = -1382.7457
Iteration 1: log pseudolikelihood = -1328.4186
Iteration 2: log pseudolikelihood = -1326.4483
Iteration 3: log pseudolikelihood = -1326.4449
Iteration 4: log pseudolikelihood = -1326.4449

Weibull regression -- log relative-hazard form

No. of subjects = 926 Number of obs = 1734
No. of failures = 808
Time at risk = 435444

Wald chi2(2) = 154.45
Log pseudolikelihood = -1326.4449 Prob > chi2 = 0.0000

(Std. Err. adjusted for 926 clusters in id)

Robust
_t Haz. Ratio Std. Err. z P>|z| [95% Conf. Interval]

x1 2.240069 .1812848 9.97 0.000 1.911504 2.625111
x2 .3206515 .0504626 -7.23 0.000 .2355458 .436507

_cons .0006962 .0001792 -28.25 0.000 .0004204 .001153

/ln_p .1771265 .0310111 5.71 0.000 .1163458 .2379071

p 1.193782 .0370205 1.123384 1.268591
1/p .8376738 .0259772 .7882759 .8901674

A one-unit change in x1 approximately doubles the hazard of failure, whereas a one-unit change
in x2 cuts the hazard to one-third its previous value. We also see that these data are close to being
exponentially distributed; p is nearly 1.

Above we mentioned that interpreting results in the PH metric is easier, though regression coefficients
are not difficult to interpret in the AFT metric. A positive coefficient means that time is decelerated
by a unit increase in the covariate in question. This may seem awkward, but think of this instead as
a unit increase in the covariate causing a delay in failure and thus increasing the expected time until
failure.

The difficulty that arises with the AFT metric is merely that it places an emphasis on log(time-to-
failure) rather than risk (hazard) of failure. With this emphasis usually comes a desire to predict the
time to failure, and therein lies the difficulty with complex survival data. Predicting the log(time to
failure) with predict assumes that the subject is at risk from time 0 until failure and has a fixed
covariate pattern over this period. With these data, such assumptions produce predictions having little
to do with the test subjects, who exhibit not only time-varying covariates but also multiple failures.

Predicting time to failure with complex survival data is difficult regardless of the metric under
which estimation took place. Those who estimate in the PH metric are probably used to dealing with
results from Cox regression, of which predicted time to failure is typically not the focus.

Example 5

The multiple-failure data above are close enough to exponentially distributed that we will reestimate
using exponential regression:

. streg x1 x2, d(exp) vce(robust)

Iteration 0: log pseudolikelihood = -1398.2504
Iteration 1: log pseudolikelihood = -1343.6083
Iteration 2: log pseudolikelihood = -1341.5932
Iteration 3: log pseudolikelihood = -1341.5893
Iteration 4: log pseudolikelihood = -1341.5893

Exponential regression -- log relative-hazard form

No. of subjects = 926 Number of obs = 1734
No. of failures = 808
Time at risk = 435444

Wald chi2(2) = 166.92
Log pseudolikelihood = -1341.5893 Prob > chi2 = 0.0000

(Std. Err. adjusted for 926 clusters in id)

Robust
_t Haz. Ratio Std. Err. z P>|z| [95% Conf. Interval]

x1 2.19065 .1684399 10.20 0.000 1.884186 2.54696
x2 .3037259 .0462489 -7.83 0.000 .2253552 .4093511

_cons .0024536 .0001535 -96.05 0.000 .0021704 .0027738

Technical note
For our “complex” survival data, we specified vce(robust)when fitting the Weibull and exponential

models. This was because these data were stset with an id() variable, and given the time-varying
covariates and multiple failures, it is important not to assume that the observations within each subject
are independent. When we specified vce(robust), it was implicit that we were “clustering” on the
groups defined by the id() variable.

You might sometimes have multiple observations per subject, which exist merely as a result
of the data-organization mechanism and are not used to record gaps, time-varying covariates, or
multiple failures. Such data could be collapsed into single-observation-per-subject data with no loss
of information. In these cases, we refer to splitting the observations to form multiple observations per
subject as noninformative. When the episode-splitting is noninformative, the model-based (nonrobust)
standard errors produced will be the same as those produced when the data are collapsed into single
records per subject. Thus, for these type of data, clustering of these multiple observations that results
from specifying vce(robust) is not critical.

Example 6

A reasonable question to ask is, “Given that we have several possible parametric models, how can
we select one?” When parametric models are nested, the likelihood-ratio or Wald test can be used
to discriminate between them. This can certainly be done for Weibull versus exponential or gamma
versus Weibull or lognormal. When models are not nested, however, these tests are inappropriate,
and the task of discriminating between models becomes more difficult. A common approach to this
problem is to use the Akaike information criterion (AIC). Akaike (1974) proposed penalizing each
log likelihood to reflect the number of parameters being estimated in a particular model and then
comparing them. Here the AIC can be defined as

AIC = −2(log likelihood) + 2(c+ p+ 1)

where c is the number of model covariates and p is the number of model-specific ancillary parameters
listed in table 1. Although the best-fitting model is the one with the largest log likelihood, the preferred
model is the one with the smallest AIC value. The AIC value may be obtained by using the estat
ic postestimation command; see [R] estat ic.

Using cancer.dta distributed with Stata, let’s first fit a generalized gamma model and test the
hypothesis that κ = 0 (test for the appropriateness of the lognormal) and then test the hypothesis that
κ = 1 (test for the appropriateness of the Weibull).

. use http://www.stata-press.com/data/r13/cancer
(Patient Survival in Drug Trial)

. stset studytime, failure(died)
(output omitted )

. replace drug = drug==2 | drug==3 // 0, placebo : 1, nonplacebo
(48 real changes made)

. streg drug age, d(gamma) nolog

failure _d: died
analysis time _t: studytime

Gamma regression -- accelerated failure-time form

No. of subjects = 48 Number of obs = 48
No. of failures = 31
Time at risk = 744

LR chi2(2) = 36.07
Log likelihood = -42.452006 Prob > chi2 = 0.0000

_t Coef. Std. Err. z P>|z| [95% Conf. Interval]

drug 1.394658 .2557198 5.45 0.000 .893456 1.895859
age -.0780416 .0227978 -3.42 0.001 -.1227245 -.0333587

_cons 6.456091 1.238457 5.21 0.000 4.02876 8.883421

/ln_sig -.3793632 .183707 -2.07 0.039 -.7394222 -.0193041
/kappa .4669252 .5419478 0.86 0.389 -.595273 1.529123

sigma .684297 .1257101 .4773897 .980881

The Wald test of the hypothesis that κ = 0 (test for the appropriateness of the lognormal) is
reported in the output above. The p-value is 0.389, suggesting that lognormal might be an adequate
model for these data.

The Wald test for κ = 1 is

. test [kappa]_cons = 1

( 1) [kappa]_cons = 1

chi2( 1) = 0.97
Prob > chi2 = 0.3253

providing some support against rejecting the Weibull model.

We now fit the exponential, Weibull, loglogistic, and lognormal models separately. To directly
compare coefficients, we will ask Stata to report the exponential and Weibull models in AFT form by
specifying the time option. The output from fitting these models and the results from the generalized
gamma model are summarized in table 2.

http://www.stata.com/manuals13/restatic.pdf#restatic

Table 2. Results obtained from streg, using cancer.dta with drug as an indicator variable

Generalized
Parameter Exponential Weibull Lognormal Loglogistic gamma

Age −.0886715 −.0714323 −.0833996 −.0803289 −.078042
Drug 1.682625 1.305563 1.445838 1.420237 1.394658
Constant 7.146218 6.289679 6.580887 6.446711 6.456091
Ancillary 1.682751 0.751136 0.429276 0.684297
Kappa 0.466925
Log likelihood −48.397094 −42.931335 −42.800864 −43.21698 −42.452006
AIC 102.7942 93.86267 93.60173 94.43396 94.90401

The largest log likelihood was obtained for the generalized gamma model; however, the lognormal
model is preferred by the AIC.

Parameterization of ancillary parameters

By default, all ancillary parameters are estimated as being constant. For example, the ancillary
parameter, p, of the Weibull distribution is assumed to be a constant that is not dependent on
any covariates. streg’s ancillary() and anc2() options allow for complete parameterization of
parametric survival models. By specifying, for example,

. streg x1 x2, d(weibull) ancillary(x2 z1 z2)

both λ and the ancillary parameter, p, are parameterized in terms of covariates.

Ancillary parameters are usually restricted to be strictly positive, in which case the logarithm of
the ancillary parameter is modeled using a linear predictor, which can assume any value on the real
line.

Example 7

Consider a dataset in which we model the time until hip fracture as Weibull for patients on the basis
of age, sex, and whether the patient wears a hip-protective device (variable protect). We believe
that the hazard is scaled according to sex and the presence of the device but believe the hazards for
both sexes to be of different shapes.

. use http://www.stata-press.com/data/r13/hip3, clear
(hip fracture study)

. streg protect age, d(weibull) ancillary(male) nolog

failure _d: fracture
analysis time _t: time1

id: id

Weibull regression -- log relative-hazard form

No. of subjects = 148 Number of obs = 206
No. of failures = 37
Time at risk = 1703

LR chi2(2) = 39.80
Log likelihood = -69.323532 Prob > chi2 = 0.0000

_t Coef. Std. Err. z P>|z| [95% Conf. Interval]

_t
protect -2.130058 .3567005 -5.97 0.000 -2.829178 -1.430938

age .0939131 .0341107 2.75 0.006 .0270573 .1607689
_cons -10.17575 2.551821 -3.99 0.000 -15.17722 -5.174269

ln_p
male -.4887189 .185608 -2.63 0.008 -.8525039 -.1249339

_cons .4540139 .1157915 3.92 0.000 .2270667 .6809611

From our estimation results, we see that ̂ln(p) = 0.454 for females and ̂ln(p) = 0.454− 0.489 =
−0.035 for males. Thus p̂ = 1.57 for females and p̂ = 0.97 for males. When we combine this with
the main equation in the model, the estimated hazards are then

ĥ(tj |xj) =

{
exp

(
−10.18− 2.13protectj + 0.09agej

)
1.57t0.57

j if female

exp
(
−10.18− 2.13protectj + 0.09agej

)
0.97t−0.03

j if male

If we believe this model, we would say that the hazard for males given age and protect is almost
constant over time.

Contrast this with what we obtain if we type

. streg protect age if male, d(weibull)

. streg protect age if !male, d(weibull)

which is completely general, because not only will the shape parameter, p, differ over both sexes,
but the regression coefficients will as well.

The anc2() option is for use only with the gamma regression model, because it contains two
ancillary parameters—anc2() is used to parametrize κ.

Stratified estimation
When we type

. streg xvars, d(distname) strata(varname)

we are asking that a completely stratified model be fit. By completely stratified, we mean that both
the model’s intercept and any ancillary parameters are allowed to vary for each level of the strata
variable. That is, we are constraining the coefficients on the covariates to be the same across strata
but allowing the intercept and ancillary parameters to vary.

Example 8

We demonstrate this by fitting a stratified lognormal model to the cancer data, with the drug
variable left in its original state: drug==1 refers to the placebo, and drug==2 and drug==3 correspond
to two alternative treatments.

. use http://www.stata-press.com/data/r13/cancer
(Patient Survival in Drug Trial)

. stset studytime, failure(died)
(output omitted )

. streg age, d(weibull) strata(drug) nolog

failure _d: died
analysis time _t: studytime

Weibull regression -- log relative-hazard form

No. of subjects = 48 Number of obs = 48
No. of failures = 31
Time at risk = 744

LR chi2(3) = 16.58
Log likelihood = -41.113074 Prob > chi2 = 0.0009

_t Coef. Std. Err. z P>|z| [95% Conf. Interval]

_t
age .1212332 .0367538 3.30 0.001 .049197 .1932694

_Sdrug_2 -4.561178 2.339448 -1.95 0.051 -9.146411 .0240556
_Sdrug_3 -3.715737 2.595986 -1.43 0.152 -8.803776 1.372302

_cons -10.36921 2.341022 -4.43 0.000 -14.95753 -5.780896

ln_p
_Sdrug_2 .4872195 .332019 1.47 0.142 -.1635257 1.137965
_Sdrug_3 .2194213 .4079989 0.54 0.591 -.5802418 1.019084

_cons .4541282 .1715663 2.65 0.008 .1178645 .7903919

Completely stratified models are fit by first generating stratum-specific indicator variables (dummy
variables) and then adding these as independent variables in the model and as covariates in the
ancillary parameter. The strata() option is thus merely a shorthand method for generating the
indicator variables from the drug categories and then placing these indicators in both the main
equation and the ancillary equation(s).

We associate the term stratification with this process by noting that the intercept term of the main
equation is a component of the baseline hazard (or baseline survivor) function. By allowing this
intercept, as well as the ancillary shape parameter, to vary with respect to the strata, we allow the
baseline functions to completely vary over the strata, analogous to a stratified Cox model.

Example 9

We can produce a less-stratified model by specifying a factor variable in the ancillary() option.

. streg age, d(weibull) ancillary(i.drug) nolog

failure _d: died
analysis time _t: studytime

Weibull regression -- log relative-hazard form

No. of subjects = 48 Number of obs = 48
No. of failures = 31
Time at risk = 744

LR chi2(1) = 9.61
Log likelihood = -44.596379 Prob > chi2 = 0.0019

_t Coef. Std. Err. z P>|z| [95% Conf. Interval]

_t
age .1126419 .0362786 3.10 0.002 .0415373 .1837466

_cons -10.95772 2.308489 -4.75 0.000 -15.48227 -6.433162

ln_p
drug

2 -.3279568 .11238 -2.92 0.004 -.5482176 -.107696
3 -.4775351 .1091141 -4.38 0.000 -.6913948 -.2636755

_cons .6684086 .1327284 5.04 0.000 .4082657 .9285514

By doing this, we are restricting not only the coefficients on the covariates to be the same across
“strata” but also the intercept, while allowing only the ancillary parameter to differ.

By using ancillary() or strata(), we may thus consider a wide variety of models, depending
on what we believe about the effect of the covariate(s) in question. For example, when fitting a
Weibull PH model to the cancer data, we may choose from many models, depending on what we
want to assume is the effect of the categorical variable drug. For all models considered below, we
assume implicitly that the effect of age is proportional on the hazard function.

1. drug has no effect:

. streg age, d(weibull)

2. The effect of drug is proportional on the hazard (scale), and the effect of age is the same for
each level of drug:

. streg age i.drug, d(weibull)

3. drug affects the shape of the hazard, and the effect of age is the same for each level of drug:

. streg age, d(weibull) ancillary(i.drug)

4. drug affects both the scale and shape of the hazard, and the effect of age is the same for each
level of drug:

. streg age, d(weibull) strata(drug)

5. drug affects both the scale and shape of the hazard, and the effect of age is different for each
level of drug:

. streg drug##c.age, d(weibull) strata(drug)

These models may be compared using Wald or likelihood-ratio tests when the models in question
are nested (such as 3 nested within 4) or by using the AIC for nonnested models.

Everything we said regarding the modeling of ancillary parameters and stratification applies to AFT
models as well, for which interpretations may be stated in terms of the baseline survivor function,
that is, the unaccelerated probability of survival past time t.

Technical note
When fitting PH models, streg will, by default, display the exponentiated regression coefficients,

labeled as hazard ratios. However, in our previous examples using ancillary() and strata(), the
regression outputs displayed the untransformed coefficients instead. This change in behavior has to
do with the modeling of the ancillary parameter. When we use one or more covariates from the main
equation to model an ancillary parameter, hazard ratios (and time ratios for AFT models) lose their
interpretation. streg, as a precaution, disallows the display of hazard/time ratios when ancillary(),
anc2(), or strata() is specified.

Keep this in mind when comparing results across various model specifications. For example, when
comparing a stratified Weibull PH model to a standard Weibull PH model, be sure that the latter is
displayed using the nohr option.

(Unshared-) frailty models

A frailty model is a survival model with unobservable heterogeneity, or frailty. At the observation
level, frailty is introduced as an unobservable multiplicative effect, α, on the hazard function, such
that

h(t|α) = αh(t)

where h(t) is a nonfrailty hazard function, say, the hazard function of any of the six parametric models
supported by streg described earlier in this entry. The frailty, α, is a random positive quantity and,
for model identifiability, is assumed to have mean 1 and variance θ.

Exploiting the relationship between the cumulative hazard function and survivor function yields
the expression for the survivor function, given the frailty

S(t|α) = exp
{
−
∫ t

0

h(u|α)du
}

= exp
{
−α

∫ t

0

f(u)
S(u)

du

}
= {S(t)}α

where S(t) is the survivor function that corresponds to h(t).

Because α is unobservable, it must be integrated out of S(t|α) to obtain the unconditional survivor
function. Let g(α) be the probability density function of α, in which case an estimable form of our
frailty model is achieved as

Sθ(t) =
∫ ∞

0

S(t|α)g(α)dα =
∫ ∞

0

{S(t)}α g(α)dα

Given the unconditional survivor function, we can obtain the unconditional hazard and density in
the usual way:

fθ(t) = − d

dt
Sθ(t) hθ(t) =

fθ(t)
Sθ(t)

Hence, an unshared-frailty model is merely a typical parametric survival model, with the addi-
tional estimation of an overdispersion parameter, θ. In a standard survival regression, the likelihood
calculations are based on S(t), h(t), and f(t). In an unshared-frailty model, the likelihood is based
analogously on Sθ(t), hθ(t), and fθ(t).

At this stage, the only missing piece is the choice of frailty distribution, g(α). In theory, any
continuous distribution supported on the positive numbers that has expectation 1 and finite variance θ is
allowed here. For mathematical tractability, however, we limit the choice to either the gamma(1/θ, θ)
distribution or the inverse-Gaussian distribution with parameters 1 and 1/θ, denoted as IG(1, 1/θ).
The gamma(a, b) distribution has probability density function

g(x) =
xa−1e−x/b

Γ(a)ba

and the IG(a, b) distribution has density

g(x) =
(

b

2πx3

)1/2

exp
{
− b

2a

(x
a
− 2 +

a

x

)}
Therefore, performing the integrations described above will show that specifying frailty(gamma)

will result in the frailty survival model (in terms of the nonfrailty survivor function, S(t))

Sθ(t) = [1− θ log {S(t)}]−1/θ

Specifying frailty(invgaussian) will give

Sθ(t) = exp
{

1
θ

(
1− [1− 2θ log {S(t)}]1/2

)}
Regardless of the choice of frailty distribution, limθ→0Sθ(t) = S(t), and thus the frailty model
reduces to S(t) when there is no heterogeneity present.

When using frailty models, distinguish between the hazard faced by the individual (subject), αh(t),
and the “average” hazard for the population, hθ(t). Similarly, an individual will have probability of
survival past time t equal to {S(t)}α, whereas Sθ(t) will measure the proportion of the population
surviving past time t. You specify S(t) as before with distribution(distname), and the list of
possible parametric forms for S(t) is given in table 1. Thus when you specify distribution() you
are specifying a model for an individual with frailty equal to one. Specifying frailty(distname)
determines which of the two above forms for Sθ(t) is used.

The output of the estimation remains unchanged from the nonfrailty version, except for the additional
estimation of θ and a likelihood-ratio test of H0: θ = 0. For more information on frailty models,
Hougaard (1986) offers an excellent introduction. For a Stata-specific overview, see Gutierrez (2002).

Example 10

Consider as an example a survival analysis of data on women with breast cancer. Our hypothetical
dataset consists of analysis times on 80 women with covariates age, smoking, and dietfat, which
measures the average weekly calories from fat (×103) in the patient’s diet over the course of the
study.



22 streg — Parametric survival models

. use http://www.stata-press.com/data/r13/bc

. list in 1/12

age smoking dietfat t dead

1. 30 1 4.919 14.2 0
2. 50 0 4.437 8.21 1
3. 47 0 5.85 5.64 1
4. 49 1 5.149 4.42 1
5. 52 1 4.363 2.81 1

6. 29 0 6.153 35 0
7. 49 1 3.82 4.57 1
8. 27 1 5.294 35 0
9. 47 0 6.102 3.74 1

10. 59 0 4.446 2.29 1

11. 35 0 6.203 15.3 0
12. 26 0 4.515 35 0

The data are well fit by a Weibull model for the distribution of survival time conditional on age,
smoking, and dietary fat. By omitting the dietfat variable from the model, we hope to introduce
unobserved heterogeneity.

. stset t, fail(dead)
(output omitted )

. streg age smoking, d(weibull) frailty(gamma)

failure _d: dead
analysis time _t: t

Fitting Weibull model:

Fitting constant-only model:

Iteration 0: log likelihood = -137.15363
Iteration 1: log likelihood = -136.3927
Iteration 2: log likelihood = -136.01557
Iteration 3: log likelihood = -136.01202
Iteration 4: log likelihood = -136.01201

Fitting full model:

Iteration 0: log likelihood = -85.933969
Iteration 1: log likelihood = -73.61173
Iteration 2: log likelihood = -68.999447
Iteration 3: log likelihood = -68.340858
Iteration 4: log likelihood = -68.136187
Iteration 5: log likelihood = -68.135804
Iteration 6: log likelihood = -68.135804

Weibull regression -- log relative-hazard form
Gamma frailty

No. of subjects = 80 Number of obs = 80
No. of failures = 58
Time at risk = 1257.07

LR chi2(2) = 135.75
Log likelihood = -68.135804 Prob > chi2 = 0.0000

_t Haz. Ratio Std. Err. z P>|z| [95% Conf. Interval]

age 1.475948 .1379987 4.16 0.000 1.228811 1.772788
smoking 2.788548 1.457031 1.96 0.050 1.00143 7.764894

_cons 4.57e-11 2.38e-10 -4.57 0.000 1.70e-15 1.23e-06

/ln_p 1.087761 .222261 4.89 0.000 .6521376 1.523385
/ln_the .3307466 .5250758 0.63 0.529 -.698383 1.359876

p 2.967622 .6595867 1.91964 4.587727
1/p .3369701 .0748953 .2179729 .520931

theta 1.392007 .7309092 .4973889 3.895711

Likelihood-ratio test of theta=0: chibar2(01) = 22.57 Prob>=chibar2 = 0.000

We could also use an inverse-Gaussian distribution to model the heterogeneity.

. streg age smoking, d(weibull) frailty(invgauss) nolog

failure _d: dead
analysis time _t: t

Weibull regression -- log relative-hazard form
Inverse-Gaussian frailty

No. of subjects = 80 Number of obs = 80
No. of failures = 58
Time at risk = 1257.07

LR chi2(2) = 125.44
Log likelihood = -73.838578 Prob > chi2 = 0.0000

_t Haz. Ratio Std. Err. z P>|z| [95% Conf. Interval]

age 1.284133 .0463256 6.93 0.000 1.196473 1.378217
smoking 2.905409 1.252785 2.47 0.013 1.247892 6.764528

_cons 1.11e-07 2.34e-07 -7.63 0.000 1.83e-09 6.79e-06

/ln_p .7173904 .1434382 5.00 0.000 .4362567 .9985241
/ln_the .2374778 .8568064 0.28 0.782 -1.441832 1.916788

p 2.049079 .2939162 1.546906 2.714273
1/p .4880241 .0700013 .3684228 .6464518

theta 1.268047 1.086471 .2364941 6.799082

Likelihood-ratio test of theta=0: chibar2(01) = 11.16 Prob>=chibar2 = 0.000

The results are similar with respect to the choice of frailty distribution, with the gamma frailty
model producing a slightly higher likelihood. Both models show a statistically significant level of
unobservable heterogeneity because the p-value for the likelihood-ratio (LR) test of H0 : θ = 0 is
virtually zero in both cases.

Technical note

With gamma-distributed or inverse-Gaussian–distributed frailty, hazard ratios decay over time in
favor of the frailty effect, and thus the displayed “Haz. Ratio” in the above output is actually the
hazard ratio only for t = 0. The degree of decay depends on θ. Should the estimated θ be close to
zero, the hazard ratios regain their usual interpretation. The rate of decay and the limiting hazard
ratio differ between the gamma and inverse-Gaussian models; see Gutierrez (2002) for details.

For this reason, many researchers prefer fitting frailty models in the AFT metric because the
interpretation of regression coefficients is unchanged by the frailty—the factors in question serve to
either accelerate or decelerate the survival experience. The only difference is that with frailty models,
the unconditional probability of survival is described by Sθ(t) rather than S(t).

Technical note
The LR test of θ = 0 is a boundary test and thus requires careful consideration concerning the

calculation of its p-value. In particular, the null distribution of the LR test statistic is not the usual χ2
1

but rather is a 50:50 mixture of a χ2
0 (point mass at zero) and a χ2

1, denoted as χ̄2
01. See Gutierrez,

Carter, and Drukker (2001) for more details.

To verify that the significant heterogeneity is caused by the omission of dietfat, we now refit
the Weibull/inverse-Gaussian frailty model with dietfat included.

. streg age smoking dietfat, d(weibull) frailty(invgauss) nolog

failure _d: dead
analysis time _t: t

Weibull regression -- log relative-hazard form
Inverse-Gaussian frailty

No. of subjects = 80 Number of obs = 80
No. of failures = 58
Time at risk = 1257.07

LR chi2(3) = 246.41
Log likelihood = -13.352142 Prob > chi2 = 0.0000

_t Haz. Ratio Std. Err. z P>|z| [95% Conf. Interval]

age 1.74928 .0985244 9.93 0.000 1.566453 1.953446
smoking 5.203549 1.70494 5.03 0.000 2.737815 9.889978
dietfat 9.229841 2.219328 9.24 0.000 5.761315 14.78655

_cons 1.07e-20 4.98e-20 -9.92 0.000 1.22e-24 9.45e-17

/ln_p 1.431742 .0978844 14.63 0.000 1.239892 1.623592
/ln_the -14.42565 2850.629 -0.01 0.996 -5601.555 5572.704

p 4.185986 .4097429 3.455242 5.071275
1/p .2388923 .0233838 .1971891 .2894153

theta 5.43e-07 .0015487 0 .

Likelihood-ratio test of theta=0: chibar2(01) = 0.00 Prob>=chibar2 = 1.000

The estimate of the frailty variance component θ is near zero, and the p-value of the test of
H0: θ = 0 equals one, indicating negligible heterogeneity. A regular Weibull model could be fit to
these data (with dietfat included), producing almost identical estimates of the hazard ratios and
ancillary parameter, p, so such an analysis is omitted here.

Also hazard ratios now regain their original interpretation. Thus an increase in weekly calories
from fat of 1,000 would increase the risk of death by more than ninefold.

Shared-frailty models

A generalization of the frailty models considered in the previous section is the shared-frailty model,
where the frailty is assumed to be group specific; this is analogous to a panel-data regression model.
For observation j from the ith group, the hazard is

hij(t|αi) = αihij(t)

for i = 1, . . . , n and j = 1, . . . , ni, where by hij(t) we mean h(t|xij), which is the individual
hazard given covariates xij .

Shared-frailty models are appropriate when you wish to model the frailties as being specific to
groups of subjects, such as subjects within families. Here a shared-frailty model may be used to
model the degree of correlation within groups; that is, the subjects within a group are correlated
because they share the same common frailty.

Example 11

Consider the data from a study of 38 kidney dialysis patients, as described in McGilchrist and
Aisbett (1991). The study is concerned with the prevalence of infection at the catheter-insertion point.
Two recurrence times (in days) are measured for each patient, and each recorded time is the time
from initial insertion (onset of risk) to infection or censoring.

. use http://www.stata-press.com/data/r13/catheter
(Kidney data, McGilchrist and Aisbett, Biometrics, 1991)

. list patient time infect age female in 1/10

patient time infect age female

1. 1 16 1 28 0
2. 1 8 1 28 0
3. 2 13 0 48 1
4. 2 23 1 48 1
5. 3 22 1 32 0

6. 3 28 1 32 0
7. 4 318 1 31.5 1
8. 4 447 1 31.5 1
9. 5 30 1 10 0

10. 5 12 1 10 0

Each patient (patient) has two recurrence times (time) recorded, with each catheter insertion
resulting in either infection (infect==1) or right-censoring (infect==0). Among the covariates
measured are age and sex (female==1 if female, female==0 if male).

One subtlety to note concerns the use of the generic term subjects. In this example, the subjects
are the individual catheter insertions, not the patients themselves. This is a function of how the data
were recorded—the onset of risk occurs at catheter insertion (of which there are two for each patient)
not, say, at the time of admission of the patient into the study. Thus we have two subjects (insertions)
within each group (patient).

It is reasonable to assume independence of patients but unreasonable to assume that recurrence
times within each patient are independent. One solution would be to fit a standard survival model,
adjusting the standard errors of the parameter estimates to account for the possible correlation by
specifying vce(cluster patient).

We could also model the correlation by assuming that the correlation is the result of a latent
patient-level effect, or frailty. That is, rather than fitting a standard model and specifying vce(cluster
patient), we fit a frailty model and specify shared(patient). Assuming that the time to infection,
given age and female, follows a Weibull distribution, and inverse-Gaussian distributed frailties, we
get

. stset time, fail(infect)
(output omitted )

. streg age female, d(weibull) frailty(invgauss) shared(patient) nolog

failure _d: infect
analysis time _t: time

Weibull regression --
log relative-hazard form Number of obs = 76
Inverse-Gaussian shared frailty Number of groups = 38

Group variable: patient

No. of subjects = 76 Obs per group: min = 2
No. of failures = 58 avg = 2
Time at risk = 7424 max = 2

LR chi2(2) = 9.84
Log likelihood = -99.093527 Prob > chi2 = 0.0073

_t Haz. Ratio Std. Err. z P>|z| [95% Conf. Interval]

age 1.006918 .013574 0.51 0.609 .9806623 1.033878
female .2331376 .1046382 -3.24 0.001 .0967322 .5618928
_cons .0110089 .0099266 -5.00 0.000 .0018803 .0644557

/ln_p .1900625 .1315342 1.44 0.148 -.0677398 .4478649
/ln_the .0357272 .7745362 0.05 0.963 -1.482336 1.55379

p 1.209325 .1590676 .9345036 1.564967
1/p .8269074 .1087666 .638991 1.070087

theta 1.036373 .8027085 .2271066 4.729362

Likelihood-ratio test of theta=0: chibar2(01) = 8.70 Prob>=chibar2 = 0.002

Contrast this with what we obtain by assuming a subject-level lognormal model:

. streg age female, d(lnormal) frailty(invgauss) shared(patient) nolog

failure _d: infect
analysis time _t: time

Lognormal regression --
accelerated failure-time form Number of obs = 76
Inverse-Gaussian shared frailty Number of groups = 38

Group variable: patient

No. of subjects = 76 Obs per group: min = 2
No. of failures = 58 avg = 2
Time at risk = 7424 max = 2

LR chi2(2) = 16.34
Log likelihood = -97.614583 Prob > chi2 = 0.0003

_t Coef. Std. Err. z P>|z| [95% Conf. Interval]

age -.0066762 .0099457 -0.67 0.502 -.0261694 .0128171
female 1.401719 .3334931 4.20 0.000 .7480844 2.055354
_cons 3.336709 .4972641 6.71 0.000 2.362089 4.311329

/ln_sig .0625872 .1256185 0.50 0.618 -.1836205 .3087949
/ln_the -1.606248 1.190775 -1.35 0.177 -3.940125 .7276282

sigma 1.064587 .1337318 .8322516 1.361783
theta .2006389 .2389159 .0194458 2.070165

Likelihood-ratio test of theta=0: chibar2(01) = 1.53 Prob>=chibar2 = 0.108

The frailty effect is insignificant at the 10% level in the latter model yet highly significant in
the former. We thus have two possible stories to tell concerning these data: If we believe the first
model, we believe that the individual hazard of infection continually rises over time (Weibull), but
there is a significant frailty effect causing the population hazard to begin falling after some time.
If we believe the second model, we believe that the individual hazard first rises and then declines
(lognormal), meaning that if a given insertion does not become infected initially, the chances that it
will become infected begin to decrease after a certain point. Because the frailty effect is insignificant,
the population hazard mirrors the individual hazard in the second model.

As a result, both models view the population hazard as rising initially and then falling past a
certain point. The second version of our story corresponds to higher log likelihood, yet perhaps not
significantly higher given the limited data. More investigation is required. One idea is to fit a more
distribution-agnostic form of a frailty model, such as a piecewise exponential (Cleves et al. 2010,
328–331) or a Cox model with frailty; see [ST] stcox.

Shared-frailty models are also appropriate when the frailties are subject specific yet there exist
multiple records per subject. Here you would share frailties across the same id() variable previously
stset. When you have subject-specific frailties and uninformative episode splitting, it makes no
difference whether you fit a shared or an unshared frailty model. The estimation results will be the
same.

http://www.stata.com/manuals13/ststcox.pdf#ststcox

Stored results
streg stores the following in e():

Scalars
e(N) number of observations
e(N sub) number of subjects
e(N fail) number of failures
e(N g) number of groups
e(k) number of parameters
e(k eq) number of equations in e(b)
e(k eq model) number of equations in overall model test
e(k aux) number of auxiliary parameters
e(k dv) number of dependent variables
e(df m) model degrees of freedom
e(ll) log likelihood
e(ll 0) log likelihood, constant-only model
e(ll c) log likelihood, comparison model
e(N clust) number of clusters
e(chi2) χ2

e(chi2 c) χ2, comparison model
e(risk) total time at risk
e(g min) smallest group size
e(g avg) average group size
e(g max) largest group size
e(theta) frailty parameter
e(aux p) ancillary parameter (weibull)
e(gamma) ancillary parameter (gompertz, loglogistic)
e(sigma) ancillary parameter (gamma, lnormal)
e(kappa) ancillary parameter (gamma)
e(p) significance
e(p c) significance, comparison model
e(rank) rank of e(V)
e(rank0) rank of e(V), constant-only model
e(ic) number of iterations
e(rc) return code
e(converged) 1 if converged, 0 otherwise

Macros
e(cmd) model or regression name
e(cmd2) streg
e(cmdline) command as typed
e(dead) d
e(depvar) t
e(title) title in estimation output
e(title2) secondary title in estimation output
e(clustvar) name of cluster variable
e(shared) frailty grouping variable
e(fr title) title in output identifying frailty
e(wtype) weight type
e(wexp) weight expression
e(t0) t0
e(vce) vcetype specified in vce()
e(vcetype) title used to label Std. Err.
e(frm2) hazard or time
e(chi2type) Wald or LR; type of model χ2 test
e(offset1) offset for main equation
e(stcurve) stcurve
e(opt) type of optimization
e(which) max or min; whether optimizer is to perform maximization or minimization
e(ml method) type of ml method
e(user) name of likelihood-evaluator program
e(technique) maximization technique
e(properties) b V
e(predict) program used to implement predict
e(predict sub) predict subprogram
e(footnote) program used to implement the footnote display
e(asbalanced) factor variables fvset as asbalanced
e(asobserved) factor variables fvset as asobserved

Matrices
e(b) coefficient vector
e(Cns) constraints matrix
e(ilog) iteration log (up to 20 iterations)
e(gradient) gradient vector
e(V) variance–covariance matrix of the estimators
e(V modelbased) model-based variance

Functions
e(sample) marks estimation sample

Methods and formulas
For an introduction to survival models, see Cleves et al. (2010). For an introduction to survival

analysis directed at social scientists, see Box-Steffensmeier and Jones (2004).

Consider for j = 1, . . . , n observations the trivariate response, (t0j , tj , dj), representing a period
of observation, (t0j , tj ], ending in either failure (dj = 1) or right-censoring (dj = 0). This structure
allows analysis of a wide variety of models and may be used to account for delayed entry, gaps,
time-varying covariates, and multiple failures per subject. Regardless of the structure of the data, once
they are stset, the data may be treated in a common manner by streg: the stset-created variable
t0 holds the t0j , t holds the tj , and d holds the dj .

For a given survivor function, S(t), the density function is obtained as

f(t) = − d

dt
S(t)

and the hazard function (the instantaneous rate of failure) is obtained as h(t) = f(t)/S(t). Available
forms for S(t) are listed in table 1. For a set of covariates from the jth observation, xj , define
Sj(t) = S(t|x = xj), and similarly define hj(t) and fj(t). For example, in a Weibull PH model,
Sj(t) = exp{− exp(xjβ)tp}.

Parameter estimation
In this command, β and the ancillary parameters are estimated via maximum likelihood. A subject

known to fail at time tj contributes to the likelihood function the value of the density at time tj
conditional on the entry time t0j , fj(tj)/Sj(t0j). A censored observation, known to survive only
up to time tj , contributes Sj(tj)/Sj(t0j), which is the probability of surviving beyond time tj
conditional on the entry time, t0j . The log likelihood is thus given by

logL =
n∑
j=1

{dj log fj(tj) + (1− dj) logSj(tj)− logSj(t0j)}

Implicit in the above log-likelihood expression are the regression parameters, β, and the ancillary
parameters because both are components of the chosen Sj(t) and its corresponding fj(t); see table 1.
streg reports maximum likelihood estimates of β and of the ancillary parameters (if any for the
chosen model). The reported log-likelihood value is logLr = logL + T , where T =

∑
log(tj) is

summed over uncensored observations. The adjustment removes the time units from logL. Whether
the adjustment is made makes no difference to any test or result since such tests and results depend
on differences in log-likelihood functions or their second derivatives, or both.

Specifying ancillary(), anc2(), or strata() will parameterize the ancillary parameter(s) by
using the linear predictor, zjαz , where the covariates, zj , need not be distinct from xj . Here streg
will report estimates of αz in addition to estimates of β. The log likelihood here is simply the log
likelihood given above, with zjαz substituted for the ancillary parameter. If the ancillary parameter
is constrained to be strictly positive, its logarithm is parameterized instead; that is, we substitute the
linear predictor for the logarithm of the ancillary parameter in the above log likelihood. The gamma
model has two ancillary parameters, σ and κ; we parameterize σ by using ancillary() and κ by
using anc2(), and the linear predictors used for each may be distinct. Specifying strata() creates
indicator variables for the strata, places these indicators in the main equation, and uses the indicators
to parameterize any ancillary parameters that exist for the chosen model.

Unshared-frailty models have a log likelihood of the above form, with Sθ(t) and fθ(t) substituted
for S(t) and f(t), respectively. Equivalently, for gamma-distributed frailties,

logL =
n∑
j=1

[
θ−1 log {1− θ logSj(t0j)} −

(
θ−1 + dj

)
log {1− θ logSj(tj)}+ dj log hj(tj)

]

and for inverse-Gaussian–distributed frailties,

logL =
n∑
j=1

[
θ−1 {1− 2θ logSj(t0j)}1/2− θ−1 {1− 2θ logSj(tj)}1/2 +

dj log hj(tj)−
1
2
dj log {1− 2θ logSj(tj)}

]
In a shared-frailty model, the frailty is common to a group of observations. Thus, to form an

unconditional likelihood, the frailties must be integrated out at the group level. The data are organized
as i = 1, . . . , n groups with the ith group comprising j = 1, . . . , ni observations. The log likelihood
is the sum of the log-likelihood contributions for each group. Define Di =

∑
j dij as the number of

failures in the ith group. For gamma frailties, the log-likelihood contribution for the ith group is

logLi =
ni∑
j=1

dij log hij(tij)− (1/θ +Di) log

1− θ
ni∑
j=1

log
Sij(tij)
Sij(t0ij)

+

Di log θ + log Γ(1/θ +Di)− log Γ(1/θ)

This formula corresponds to the log-likelihood contribution for multiple-record data. For single-record
data, the denominator Sij(t0ij) is equal to 1. This formula is not applicable to data with delayed
entries or gaps.

For inverse-Gaussian frailties, define

Ci =

1− 2θ
ni∑
j=1

log
Sij(tij)
Sij(t0ij)


−1/2

The log-likelihood contribution for the ith group then becomes

logLi = θ−1(1− C−1
i ) +B(θCi, Di) +

ni∑
j=1

dij {log hij(tij) + logCi}

The function B(a, b) is related to the modified Bessel function of the third kind, commonly known
as the BesselK function; see Wolfram (1999, 767). In particular,

B(a, b) = a−1 +
1
2

{
log
(

2
π

)
− log a

}
+ log BesselK

(
1
2
− b, a−1

)
For both unshared- and shared-frailty models, estimation of θ takes place jointly with the estimation
of β and the ancillary parameters.

This command supports the Huber/White/sandwich estimator of the variance and its clustered
version using vce(robust) and vce(cluster clustvar), respectively. See [P] robust, particularly
Maximum likelihood estimators and Methods and formulas. If observations in the dataset represent
repeated observations on the same subjects (that is, there are time-varying covariates), the assumption
of independence of the observations is highly questionable, meaning that the conventional estimate
of variance is not appropriate. We strongly advise that you use the vce(robust) and vce(cluster
clustvar) options here. (streg knows to specify vce(cluster clustvar) if you specify vce(robust).)
vce(robust) and vce(cluster clustvar) do not apply in shared-frailty models, where the correlation
within groups is instead modeled directly.

streg also supports estimation with survey data. For details on VCEs with survey data, see
[SVY] variance estimation.

http://www.stata.com/manuals13/p_robust.pdf#p_robust
http://www.stata.com/manuals13/p_robust.pdf#p_robustRemarksandexamplesMaximumlikelihoodestimators
http://www.stata.com/manuals13/p_robust.pdf#p_robustMethodsandformulas
http://www.stata.com/manuals13/svyvarianceestimation.pdf#svyvarianceestimation

Benjamin Gompertz (1779–1865) came from a Jewish family who left Holland and settled in
England. He was self-educated in mathematics and in 1824 was appointed as actuary and head
clerk of the Alliance Assurance Company. He carried out pioneer work on the application of
differential calculus to actuarial questions, particularly the dependence of mortality on age, and
contributed to astronomy and the study of scientific instruments.

Ernst Hjalmar Waloddi Weibull (1887–1979) was a Swedish applied physicist most famous for his
work on the statistics of material properties. He worked in Germany and Sweden as an inventor
and a consulting engineer, publishing his first paper on the propagation of explosive waves in
1914, thereafter becoming a full professor at the Royal Institute of Technology in 1924. Weibull
wrote two important papers, “Investigations into strength properties of brittle materials” and “The
phenomenon of rupture in solids”, which discussed his ideas about the statistical distributions of
material strength. These articles came to the attention of engineers in the late 1930s.� �

Also see
[ST] streg postestimation — Postestimation tools for streg

[ST] stcurve — Plot survivor, hazard, cumulative hazard, or cumulative incidence function

[ST] stcox — Cox proportional hazards model

[ST] stcrreg — Competing-risks regression

[ST] sts — Generate, graph, list, and test the survivor and cumulative hazard functions

[ST] stset — Declare data to be survival-time data

[MI] estimation — Estimation commands for use with mi estimate

[SVY] svy estimation — Estimation commands for survey data

[U] 20 Estimation and postestimation commands

